
NAG Library Routine Document

G02HMF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

G02HMF computes a robust estimate of the covariance matrix for user-supplied weight functions. The
derivatives of the weight functions are not required.

2 Specification

SUBROUTINE G02HMF (UCV, RUSER, INDM, N, M, X, LDX, COV, A, WT, THETA,
BL, BD, MAXIT, NITMON, TOL, NIT, WK, IFAIL)

&

INTEGER INDM, N, M, LDX, MAXIT, NITMON, NIT, IFAIL
REAL (KIND=nag_wp) RUSER(*), X(LDX,M), COV(M*(M+1)/2), A(M*(M+1)/2),

WT(N), THETA(M), BL, BD, TOL, WK(2*M)
&

EXTERNAL UCV

3 Description

For a set of n observations on m variables in a matrix X, a robust estimate of the covariance matrix, C,
and a robust estimate of location, �, are given by

C ¼ �2 ATA
� ��1

;

where �2 is a correction factor and A is a lower triangular matrix found as the solution to the following
equations.

zi ¼ A xi � �ð Þ
1

n

Xn
i¼1

w zik k2
� �

zi ¼ 0

and

1

n

Xn
i¼1

u zik k2
� �

ziz
T
i � v zik k2

� �
I ¼ 0;

where xi is a vector of length m containing the elements of the ith row of X,

zi is a vector of length m,

I is the identity matrix and 0 is the zero matrix.

and w and u are suitable functions.

G02HMF covers two situations:

(i) v tð Þ ¼ 1 for all t,

(ii) v tð Þ ¼ u tð Þ.
The robust covariance matrix may be calculated from a weighted sum of squares and cross-products
matrix about � using weights wti ¼ u zik kð Þ. In case (i) a divisor of n is used and in case (ii) a divisor

of
Xn
i¼1

wti is used. If w :ð Þ ¼ ffiffiffiffiffiffiffiffi
u :ð Þp

, then the robust covariance matrix can be calculated by scaling each

row of X by
ffiffiffiffiffiffiffi
wti

p
and calculating an unweighted covariance matrix about �.

In order to make the estimate asymptotically unbiased under a Normal model a correction factor, �2, is
needed. The value of the correction factor will depend on the functions employed (see Huber (1981)
and Marazzi (1987)).

G02HMF finds A using the iterative procedure as given by Huber; see Huber (1981).

Ak ¼ Sk þ Ið ÞAk�1

and

�jk ¼
bj
D1

þ �jk�1 ;

where Sk ¼ sjl
� �

, for j ¼ 1; 2; . . . ;m and l ¼ 1; 2; . . . ;m is a lower triangular matrix such that

sjl ¼
�min max hjl=D2;�BL

� �
;BL

� �
; j > l

�min max 1
2 hjj=D2 � 1
� �

;�BD
� �

;BD
� �

; j ¼ l

8<
: ;

where

D1 ¼
Xn
i¼1

w zik k2
� �

D2 ¼
Xn
i¼1

u zik k2
� �

hjl ¼
Xn
i¼1

u zik k2
� �

zijzil, for j � l

bj ¼
Xn
i¼1

w zik k2
� �

xij � bj
� �

and BD and BL are suitable bounds.

The value of � may be chosen so that C is unbiased if the observations are from a given distribution.

G02HMF is based on routines in ROBETH; see Marazzi (1987).

4 References

Huber P J (1981) Robust Statistics Wiley

Marazzi A (1987) Weights for bounded influence regression in ROBETH Cah. Rech. Doc. IUMSP, No.
3 ROB 3 Institut Universitaire de Médecine Sociale et Préventive, Lausanne

5 Arguments

1: UCV – SUBROUTINE, supplied by the user. External Procedure

UCV must return the values of the functions u and w for a given value of its argument.

The specification of UCV is:

SUBROUTINE UCV (T, RUSER, U, W)

REAL (KIND=nag_wp) T, RUSER(*), U, W

1: T – REAL (KIND=nag_wp) Input

On entry: the argument for which the functions u and w must be evaluated.

2: RUSERð�Þ – REAL (KIND=nag_wp) array User Workspace

UCV is called with the argument RUSER as supplied to G02HMF. You should use the
array RUSER to supply information to UCV.

3: U – REAL (KIND=nag_wp) Output

On exit: the value of the u function at the point T.

Constraint: U � 0:0.

4: W – REAL (KIND=nag_wp) Output

On exit: the value of the w function at the point T.

Constraint: W � 0:0.

UCV must either be a module subprogram USEd by, or declared as EXTERNAL in, the (sub)
program from which G02HMF is called. Arguments denoted as Input must not be changed by
this procedure.

2: RUSERð�Þ – REAL (KIND=nag_wp) array User Workspace

RUSER is not used by G02HMF, but is passed directly to UCV and should be used to pass
information to this routine.

3: INDM – INTEGER Input

On entry: indicates which form of the function v will be used.

INDM ¼ 1
v ¼ 1.

INDM 6¼ 1
v ¼ u.

4: N – INTEGER Input

On entry: n, the number of observations.

Constraint: N > 1.

5: M – INTEGER Input

On entry: m, the number of columns of the matrix X, i.e., number of independent variables.

Constraint: 1 � M � N.

6: XðLDX;MÞ – REAL (KIND=nag_wp) array Input

On entry: Xði; jÞ must contain the ith observation on the jth variable, for i ¼ 1; 2; . . . ; n and
j ¼ 1; 2; . . . ;m.

7: LDX – INTEGER Input

On entry: the first dimension of the array X as declared in the (sub)program from which
G02HMF is called.

Constraint: LDX � N.

8: COVðM� Mþ 1ð Þ=2Þ – REAL (KIND=nag_wp) array Output

On exit: a robust estimate of the covariance matrix, C. The upper triangular part of the matrix C
is stored packed by columns (lower triangular stored by rows), that is Cij is returned in
COVðj� j� 1ð Þ=2þ iÞ, i � j.

9: AðM� Mþ 1ð Þ=2Þ – REAL (KIND=nag_wp) array Input/Output

On entry: an initial estimate of the lower triangular real matrix A. Only the lower triangular
elements must be given and these should be stored row-wise in the array.

The diagonal elements must be 6¼ 0, and in practice will usually be > 0. If the magnitudes of
the columns of X are of the same order, the identity matrix will often provide a suitable initial
value for A. If the columns of X are of different magnitudes, the diagonal elements of the initial
value of A should be approximately inversely proportional to the magnitude of the columns of X.

Constraint: Aðj � j � 1ð Þ=2þ jÞ 6¼ 0:0, for j ¼ 1; 2; . . . ;m.

On exit: the lower triangular elements of the inverse of the matrix A, stored row-wise.

10: WTðNÞ – REAL (KIND=nag_wp) array Output

On exit: WTðiÞ contains the weights, wti ¼ u zik k2
� �

, for i ¼ 1; 2; . . . ; n.

11: THETAðMÞ – REAL (KIND=nag_wp) array Input/Output

On entry: an initial estimate of the location argument, �j , for j ¼ 1; 2; . . . ;m.

In many cases an initial estimate of �j ¼ 0, for j ¼ 1; 2; . . . ;m, will be adequate. Alternatively
medians may be used as given by G07DAF.

On exit: contains the robust estimate of the location argument, �j , for j ¼ 1; 2; . . . ;m.

12: BL – REAL (KIND=nag_wp) Input

On entry: the magnitude of the bound for the off-diagonal elements of Sk, BL.

Suggested value: BL ¼ 0:9.

Constraint: BL > 0:0.

13: BD – REAL (KIND=nag_wp) Input

On entry: the magnitude of the bound for the diagonal elements of Sk, BD.

Suggested value: BD ¼ 0:9.

Constraint: BD > 0:0.

14: MAXIT – INTEGER Input

On entry: the maximum number of iterations that will be used during the calculation of A.

Suggested value: MAXIT ¼ 150.

Constraint: MAXIT > 0.

15: NITMON – INTEGER Input

On entry: indicates the amount of information on the iteration that is printed.

NITMON > 0
The value of A, � and � (see Section 7) will be printed at the first and every NITMON
iterations.

NITMON � 0
No iteration monitoring is printed.

When printing occurs the output is directed to the current advisory message channel (See
X04ABF.)

16: TOL – REAL (KIND=nag_wp) Input

On entry: the relative precision for the final estimate of the covariance matrix. Iteration will stop
when maximum � (see Section 7) is less than TOL.

Constraint: TOL > 0:0.

17: NIT – INTEGER Output

On exit: the number of iterations performed.

18: WKð2�MÞ – REAL (KIND=nag_wp) array Workspace

19: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this argument you should
refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, if you are not familiar with this argument, the
recommended value is 0. When the value �1 or 1 is used it is essential to test the value of
IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL ¼ 1

On entry, N � 1,
or M < 1,
or N < M,
or LDX < N.

IFAIL ¼ 2

On entry, TOL � 0:0,
or MAXIT � 0,
or diagonal element of A ¼ 0:0,
or BL � 0:0,
or BD � 0:0.

IFAIL ¼ 3

A column of X has a constant value.

IFAIL ¼ 4

Value of U or W returned by UCV < 0.

IFAIL ¼ 5

The routine has failed to converge in MAXIT iterations.

IFAIL ¼ 6

Either the sum D1 or the sum D2 is zero. This may be caused by the functions u or w being too
strict for the current estimate of A (or C). You should either try a larger initial estimate of A or
make the u and w functions less strict.

IFAIL ¼ �99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

On successful exit the accuracy of the results is related to the value of TOL; see Section 5. At an
iteration let

(i) d1 ¼ the maximum value of sjl
�� ��

(ii) d2 ¼ the maximum absolute change in wt ið Þ
(iii) d3 ¼ the maximum absolute relative change in �j

and let � ¼ max d1; d2; d3ð Þ. Then the iterative procedure is assumed to have converged when � < TOL.

8 Parallelism and Performance

G02HMF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor
library used by this implementation. Consult the documentation for the vendor library for further
information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the
OpenMP environment used within this routine. Please also consult the Users' Note for your
implementation for any additional implementation-specific information.

9 Further Comments

The existence of A will depend upon the function u (see Marazzi (1987)); also if X is not of full rank a
value of A will not be found. If the columns of X are almost linearly related, then convergence will be
slow.

If derivatives of the u and w functions are available then the method used in G02HLF will usually give
much faster convergence.

10 Example

A sample of 10 observations on three variables is read in along with initial values for A and � and
argument values for the u and w functions, cu and cw. The covariance matrix computed by G02HMF is
printed along with the robust estimate of �.

UCV computes the Huber's weight functions:

u tð Þ ¼ 1; if t � c2u

u tð Þ ¼ cu
t2
; if t > c2u

and

w tð Þ ¼ 1; if t � cw

w tð Þ ¼ cw
t
; if t > cw:

10.1 Program Text

! G02HMF Example Program Text
! Mark 26 Release. NAG Copyright 2016.

Module g02hmfe_mod

! G02HMF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Accessibility Statements ..
Private
Public :: ucv

! .. Parameters ..
Integer, Parameter, Public :: iset = 1, nin = 5, nout = 6

Contains
Subroutine ucv(t,ruser,u,w)

! u function

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: t
Real (Kind=nag_wp), Intent (Out) :: u, w

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Inout) :: ruser(*)

! .. Local Scalars ..
Real (Kind=nag_wp) :: cu, cw, t2

! .. Executable Statements ..
cu = ruser(1)
u = 1.0_nag_wp
If (t/=0.0_nag_wp) Then

t2 = t*t
If (t2>cu) Then

u = cu/t2
End If

End If
! w function

cw = ruser(2)
If (t>cw) Then

w = cw/t
Else

w = 1.0_nag_wp
End If

End Subroutine ucv
End Module g02hmfe_mod
Program g02hmfe

! G02HMF Example Main Program

! .. Use Statements ..
Use nag_library, Only: g02hmf, nag_wp, x04abf, x04ccf
Use g02hmfe_mod, Only: iset, nin, nout, ucv

! .. Implicit None Statement ..
Implicit None

! .. Local Scalars ..
Real (Kind=nag_wp) :: bd, bl, tol
Integer :: i, ifail, indm, la, lcov, ldx, &

lruser, m, maxit, n, nadv, nit, &
nitmon

! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: a(:), cov(:), ruser(:), theta(:), &

wk(:), wt(:), x(:,:)
! .. Executable Statements ..

Write (nout,*) ’G02HMF Example Program Results’
Write (nout,*)

! Skip heading in data file
Read (nin,*)

! Read in the problem size
Read (nin,*) n, m

ldx = n
lruser = 2
la = ((m+1)*m)/2
lcov = la
Allocate (x(ldx,m),ruser(lruser),cov(lcov),a(la),wt(n),theta(m),wk(2*m))

! Read in data
Read (nin,*)(x(i,1:m),i=1,n)

! Read in the initial value of A
Read (nin,*) a(1:la)

! Read in the initial value of THETA
Read (nin,*) theta(1:m)

! Read in the values of the parameters of the ucv functions
Read (nin,*) ruser(1:lruser)

! Read in the control parameters
Read (nin,*) indm, nitmon, bl, bd, maxit, tol

! Set the advisory channel to NOUT for monitoring information
If (nitmon/=0) Then

nadv = nout
Call x04abf(iset,nadv)

End If

! Compute robust estimate of variance / covariance matrix
ifail = 0
Call g02hmf(ucv,ruser,indm,n,m,x,ldx,cov,a,wt,theta,bl,bd,maxit,nitmon, &

tol,nit,wk,ifail)

! Display results
Write (nout,99999) ’G02HMF required ’, nit, ’ iterations to converge’
Write (nout,*)
Flush (nout)
ifail = 0
Call x04ccf(’Upper’,’Non-Unit’,m,cov,’Robust covariance matrix’,ifail)
Write (nout,*)
Write (nout,*) ’Robust estimates of THETA’
Write (nout,99998) theta(1:m)

99999 Format (1X,A,I0,A)
99998 Format (1X,F10.3)

End Program g02hmfe

10.2 Program Data

G02HMF Example Program Data
10 3 : N,M

3.4 6.9 12.2
6.4 2.5 15.1
4.9 5.5 14.2
7.3 1.9 18.2
8.8 3.6 11.7
8.4 1.3 17.9
5.3 3.1 15.0
2.7 8.1 7.7
6.1 3.0 21.9
5.3 2.2 13.9 : End of X
1.0 0.0 1.0 0.0 0.0 1.0 : A
0.0 0.0 0.0 : THETA
4.0 2.0 : RUSER

1 0 0.9 0.9 50 5.0E-5 : INDM,NITMON,BL,BD,MAXIT,TOL

10.3 Program Results

G02HMF Example Program Results

G02HMF required 34 iterations to converge

Robust covariance matrix
1 2 3

1 3.2779 -3.6918 4.7391
2 5.2841 -6.4087
3 11.8373

Robust estimates of THETA
5.700
3.864

14.704

